// @version=5

//dichiarazione variabili utilizzate per aprire e chiudere le transazioni
var spread = 30                                     //spread applicato ad ogni transazione (espresso in pip)
var valorePip = 0.010000                            //valore di 1 pip per 1 contratto (espresso in $)
var delta_TPSL = spread*valorePip*100               //valore utilizzato per ottimizzare il calcolo di TP e SL
var commissioni = spread*valorePip/2                //commissione applicata ad ogni transazione per 1 contratto (espressa in $)
var contratto = 1                                   //numero di contratti acquistati ad ogni transazione
var float TP = 10000+delta_TPSL                     //TP espresso in 0,1 pip. Il profitto è pari a circa al 100,00% del capitale investito per aprire una transazione, compresa la leva
var float SL = 1000-delta_TPSL                      //SL espresso in 0,1 pip. La perdita è pari a circa al 10,00% del capitale investito per aprire una transazione, compresa la leva

//dichiarazione strategia: Nome strategia, Mostra sul grafico, Valuta in Dollari, Contratti acquistati ad ogni transazione, Commissione applicata per ogni contratto acquistato
strategy("Pine Script - Strategy", overlay = true, currency = "USD", default_qty_type = strategy.fixed, default_qty_value = contratto, commission_type = strategy.commission.cash_per_contract, commission_value = commissioni)

//verifica che non sia già stato modificato lo SL nella precedente candela
if SL[1]<(1000-delta_TPSL)
    TP := TP[1]
    SL := SL[1]

//aumento dello SL se la transazione in corso raggiunge o supera un determinato profitto
for i = 100 to 10 by 5
    if (SL>((i*-100)+500-delta_TPSL) and (strategy.openprofit/contratto)>i)
        SL := (i*-100)+500-delta_TPSL
        break
if SL[1]>SL
    strategy.exit("ExitAcq", "Acq", profit = TP, loss = SL)
    strategy.exit("ExitVen", "Ven", profit = TP, loss = SL)

//calcolo degli indicatori
ema = ta.ema(close, 21)
[middleBand, upperBand, lowerBand] = ta.bb(close, 20, 2)
change = ta.change(close, 14)
atr = ta.atr(14)
[macdLine, signalLine, _] = ta.macd(close, 12, 26, 9)
rsiValue = ta.rsi(close, 14)
stochD = ta.stoch(close, high, low, 14)

//calcolo delle variabili per l'apertura di posizioni long
longcondition1 = (macdLine>0.00 and rsiValue<30 and stochD<20)
//apertura posizioni long se si verificano le condizioni di cui sopra (senza transazione in corso)
if (strategy.position_size==0 and longcondition1)
    SL := 500-delta_TPSL
    strategy.entry("Acq", strategy.long)
    strategy.exit("ExitAcq", "Acq", profit = TP, loss = SL)

//calcolo delle variabili per l'apertura di posizioni short
shortcondition1 = (macdLine<0.00 and rsiValue>70 and stochD>80)
//apertura posizioni short se si verificano le condizioni di cui sopra (senza transazione in corso)
if (strategy.position_size==0 and shortcondition1)
    SL := 500-delta_TPSL
    strategy.entry("Ven", strategy.short)
    strategy.exit("ExitVen", "Ven", profit = TP, loss = SL)

//calcolo delle variabili per l'apertura di posizioni long
longcondition2 = (open-close<0.00 and open<ema and close>ema and open<upperBand and close>upperBand and change>atr and macdLine>1.00 and macdLine>signalLine)
//apertura posizioni long se si verificano le condizioni di cui sopra (senza transazione in corso)
if (strategy.position_size==0 and longcondition2)
    SL := 1000-delta_TPSL
    strategy.entry("Acq", strategy.long)
    strategy.exit("ExitAcq", "Acq", profit = TP, loss = SL)

//calcolo delle variabili per l'apertura di posizioni short
shortcondition2 = (open-close>0.00 and open>ema and close<ema and open>lowerBand and close<lowerBand and change<(atr*-1) and macdLine<-1.00 and macdLine<signalLine)
//apertura posizioni short se si verificano le condizioni di cui sopra (senza transazione in corso)
if (strategy.position_size==0 and shortcondition2)
    SL := 1000-delta_TPSL
    strategy.entry("Ven", strategy.short)
    strategy.exit("ExitVen", "Ven", profit = TP, loss = SL)

